

Equilibrium of Particles

3.1E Parametric Tables

The examples in Chapters 1 and 2 show how to use EES to obtain a single solution to a set of equations. In engineering design problems, we often want to run a parametric study in which the effect of one variable (sometimes called the independent variable) on another variable (the dependent variable) is examined. This type of study is accomplished in EES using a Parametric Table.

In order to provide some context for this discussion let's return to Example E2.1 which determined the resultant force on the eyebolt shown in Figure 3.1.




Figure 3.1. Eyebolt from Example E2.1.

The Equations and Solution Windows are shown in Figure 3.2 and provides the magnitude and angle of the resultant force.




Figure 3.2. (a) Equations Window and (b) Solutions Window for Example E2.1.

Creating a Parametric Table

We might want to examine the magnitude and angle of the resultant force as we change the loading condition. For example, let's vary the magnitude of force \vec{F}_1 . To do this, first we need to create a variable, magF_1, that can be assigned and used to set the magnitude of the force.

```
magF_1 = 200 [lbf]           "magnitude of force 1"  
F_1 = VectorAssignPolar(magF_1, 60 [deg])  "force 1 on eyebolt"
```

Next we need to create a Parametric Table. Select New Parametric Table from the Tables menu. The New Parametric Table Dialog will appear, as shown in Figure 3.3.




Figure 3.3. New Parametric Table Dialog.

The window on the left side of the dialog provides a list of all of the variables that are included in the Equations Window. In this example the list includes each component of every vector variable that has been defined as well as the scalar variables. Highlight the independent and dependent variables of interest (in this case, the variables magF_1, magR, and angleR) by clicking on them. Then select Add in order to add these variables to the list on the right side of the dialog, which shows the variables to be included in the table. Select OK in order to create the Parametric Table, shown in Figure 3.4. By default, there are 10 runs (rows) in the table. Runs can be added or removed by selecting Insert/Delete Runs from the Tables menu or by right-clicking on any run number and selecting Insert Runs or Delete Runs. There is a column for each of the variables included in the table. Columns can be added or deleted by selecting Insert/Delete Vars from the Tables menu or right clicking on any variable and selecting Insert Column or Delete.

1..10	angleR [deg]	magF ₁ [lbf]	magR [lbf]
Run 1			
Run 2			
Run 3			
Run 4			
Run 5			
Run 6			
Run 7			
Run 8			
Run 9			
Run 10			

Figure 3.4. Empty Parametric Table.

Alter Values

To carry out a parametric study in which the magnitude of force 1 is changed, it is necessary to fill in the column for the variable magF_1 with force values that are of interest. Right-click on the column header and select Alter Values (or click on the triangular icon in the column header) to bring up the Alter Values Dialog shown in Figure 3.5.

magF_1: Column 2

First Row: 1

Last Row: 10

Enter Values

First Value: 100 lbf

Last (linear): 1000 lbf

Repeat pattern every: 10 rows

Buttons: Apply, OK, Cancel

Figure 3.5. Alter Values Dialog.

Select the rows to be filled in (rows 1 to 10, for this example) and the pattern to be used to enter the values. In Figure 3.5, the dialog is filled in so that the magnitude of the force varies from 100 lb_f (in row 1) to 1000 lb_f (in row 10) in equally spaced intervals. The result is shown in Figure 3.6.

1..10	angleR [deg]	magF ₁ [lbf]	magR [lbf]
Run 1		100	
Run 2		200	
Run 3		300	
Run 4		400	
Run 5		500	
Run 6		600	
Run 7		700	
Run 8		800	
Run 9		900	
Run 10		1000	

Figure 3.6. Parametric Table with values of the variable magF_1 set.

Solving a Parametric Table

The Parametric Table is solved and filled in using the Solve Table command from the Calculate menu (or using the shortcut key F3). EES will begin with the first run that is specified by the Solve Table command and look in the table to see which columns in the corresponding row have specified values. It will then specify the value of these variables in the Equations Window, solve the resulting system of equations, and fill in the values of the remaining columns based on the solution. Select Solve Table to bring up the Solve Table dialog shown in Figure 3.7.

Figure 3.7. Solve Table Dialog.

The Solve Table dialog is initially set so that EES will start with run 1 and end with run 10. By pressing OK, EES will go to run 1 of the Parametric Table shown in Figure 3.6 and find that the value of the variable magF₁ is set to 100 lbf. Therefore, you can imagine that EES places the following equation in the Equations Window.

$$\text{magF}_1 = 100 \text{ [lbf]}$$

Of course, this presents a problem because the statement:

`magF_1 = 200 [lbf]` "magnitude of force 1"

already exists in the Equations Window. The variable `magF_1` cannot be set twice as it causes the equation set to be over-constrained. Solving the table will result in the error dialog shown in Figure 3.8.





Figure 3.8. Error message that results from over-constraining the equation set with the Parametric Table.

This problem can be alleviated by commenting out the equation that specifies the magnitude of force 1 in the Equations Window.

`{magF_1 = 200 [lbf]}` "magnitude of force 1"

Now select Solve Table from the Calculate menu and EES will fill in each column of the Parametric Table as shown in Figure 3.9.



The image shows a screenshot of the "Parametric Table" window in EES. The window title is "Parametric Table" and it has standard window controls. Below the title bar, it says "Table 1". There are three columns with dropdown arrows: "1" (containing a green play button and "1..10"), "2" (containing "angleR [deg]"), and "3" (containing "magF₁ [lbf]"). To the right of these are two more columns: "magR [lbf]". The table contains 10 rows of data, labeled "Run 1" through "Run 10".

	1	2	3	
	1..10	angleR [deg]	magF ₁ [lbf]	magR [lbf]
Run 1		161.5	100	906.6
Run 2		155.2	200	892
Run 3		148.8	300	888.5
Run 4		142.4	400	896.3
Run 5		136.2	500	914.9
Run 6		130.3	600	943.8
Run 7		124.8	700	982.1
Run 8		119.7	800	1029
Run 9		115.1	900	1083
Run 10		111	1000	1143

Figure 3.9. Solved Parametric Table.

The \$If, \$IfNot, \$Else and \$EndIf Directive Statements

In order to run the Parametric Table we had to comment out the line in the Equations Window that specified the value of the variable magF_1. This change temporarily removes the line of code; it can be returned by highlighting the text, right clicking, and selecting Undo Comment {} from the pop-up menu.

```
magF_1 = 200 [lbf]           "magnitude of force 1"
```

A more elegant method for removing one or more lines of code when a Parametric Table is being solved is to use the \$If directive. The \$If directive is used according to:

```
$If Condition
  line(s) of code to be executed if Condition is true
$Else
  line(s) of code to be executed if Condition is false
$EndIf
```

The \$IfNot directive is used according to:

```
$IfNot Condition
  line(s) of code to be executed if Condition is not true
$Else
  line(s) of code to be executed if Condition is true
$EndIf
```

In each case, **Condition** is a keyword that indicates an execution condition. There are many such keywords recognized by EES but the one relevant to this discussion is **ParametricTable**, which evaluates to True when the equations are being solved from a Parametric Table. For our problem, we want to specify the value of the variable magF_1 in the Equations Window only if **Parametric** is false:

```
$IfNot Parametric
  magF_1 = 200 [lbf]           "magnitude of force 1"
$EndIf
```

You should find that your equation set now runs if you select either Solve or Solve Table from the Calculate menu (or if you press either F2 or F3, respectively).

3.2E Plotting Data in EES

Section 3.1E describes how to create a Parametric Table that includes the values of one or more dependent variables calculated over a range of values of an independent variable. This information is viewed most conveniently in the form of a plot. In this section we will discuss how to create a basic plot in order to show the data contained in the Parametric Table created in Section 3.1E.

Generating a Plot

To generate a plot, select New Plot Window from the Plots menu in order to access the New Plot Setup dialog, shown in Figure 3.10. The upper right portion of the dialog is used to select the source of the data; here we will use the Parametric Table named Table 1. The two list boxes in the dialog allow you to specify the independent (x -axis) and dependent (y -axis) data. Figure 3.10 shows the New Plot Dialog set up to plot the magnitude of the resultant force from Example E2.1 (magR) as a function of the magnitude of the applied force 1 (magF_1).




Figure 3.10. New Plot Setup Dialog.

Select OK to create the plot, which is shown in Figure 3.11.




Figure 3.11. Plot showing the magnitude of the resultant force as a function of the magnitude of the applied force 1.

Modifying the Axes

Almost every aspect of the plot can be modified in order to customize or improve it. Double-click (or right-click) on either axis label to bring up the Format Text Item Dialog. You can change the axis label to something more descriptive than the variable name and include units, change the font, etc. The axis scale can be adjusted by placing the mouse over any of the axes (left, right, bottom, or top) and clicking the right mouse button. This action will bring up the Modify Axis Dialog shown in Figure 3.12, which allows you to make adjustments to the axis scale, add grid lines, etc.




Figure 3.12. Modify Axis Dialog.

An improved version of the plot is shown in Figure 3.13. Note that by selecting Copy Plot from the Edit menu, a .bmp or .emf version of the plot will be placed in the clipboard, allowing you to paste it into other applications.





Figure 3.13. Improved version of the plot shown in Figure 3.11.

Overlaying Plots

Multiple data series can be overlaid onto the same plot. For example, we may want to study how the magnitude of the resultant force varies with F_1 as we apply different values of F_2 . Let's change the magnitude of applied force 2 from 500 lb_f to 600 lb_f

```
F 2 = 600 [lbf]*VectorAssign((-1/sqrt(10)),3/sqrt(10),0) "force 2"
```

and then run the Parametric Table again. Select Overlay Plot from the Plots menu in order to plot the magnitude of the resultant force as a function of the magnitude of force 1 for the adjusted value of F_2 . Your plot should now have two sets of data. You can select the Add Text item from the Plot Tool bar (), to add either labels or a legend to the plot. The result is shown in Figure 3.14.




Figure 3.14. Plot with two data series included.

Modifying Plots

Double click (or right click) anywhere on the plot in order to access the Modify Plot dialog shown in Figure 3.15. The upper window lists all of the data series that appear in the plot. You can delete one or more of these series by selecting Delete button. The characteristics of each plotted data series (e.g., the line thickness, color, symbols, etc.) can be adjusted.




Figure 3.15. Modify Plot Dialog.

At the bottom left of the Modify Plot dialog, controls are provided to allow data to be placed on the primary (X1 and Y1) or secondary axes (X2 and Y2). The primary axes are shown at the bottom (for the x -axis) and left (for the y -axis) of the plot. The secondary axes are shown at the top (for x -axis) and right (for the y -axis). For example, Figure 3.16 shows the magnitude and angle of the resultant force as a function of the magnitude of applied force 1. Because the scale of these two values is so different, the plot is much more readable when the angle is plotted using a secondary y -axis.




Figure 3.16. The magnitude and angle of the resultant force as a function of the magnitude of applied force 1. The angle of the resultant force is shown in a secondary y-axis.

The Automatic update option in the Modify Plot dialog causes the data series to be re-plotted each time the data source changes (e.g., each time the data in the Parametric Table are adjusted). This option is useful if you want to adjust parameters in your model and immediately see how they affect a plotted result.

EXAMPLE E3.1



Reactions and Force Polygon

We will revisit Example 3.4 from the text. The structure consists of a collar at B that is free to slide along a straight fixed bar AC . Mounted on the collar is a frictionless pulley, around which a cable supporting a 5 lb weight is wrapped. The collar is further supported by a bar BD .




Figure 1. Problem from Example 3.4 in Gray et al. (2023).

- If $\alpha = 0^\circ$, determine the force in bar BD needed to keep the system in equilibrium.
- Determine the value of α that will provide for the smallest force in bar BD , and determine the value of this force.

SOLUTION

Road Map We will analyze the problem using a free body diagram on the collar making sure that the angle of the bar, α , is a variable that can be adjusted in order to carry out a parametric study in part (b). The free body diagram consists of forces from the cable which have prescribed directions and a reaction force from the bar BC on the collar which can have no component in the direction of the bar due to its frictionless nature. Finally, there is a reaction force from the bar BD which has a direction specified by the angle α .

Part (a)

Governing Equations The problem can be done either using vector or scalar variables and we should get the same answer. Initially let's solve without using vectors to clearly illustrate the summation of the scalar components of the forces in each direction. A free body diagram on the weight provides the equation

$$T = W \quad (1)$$

where T is the tension in the cable. A free body diagram on the collar and pulley is shown in Figure 2 and includes the two forces from the cable as well as the reaction force on the collar from rod AC (R) and the force from bar BD (F_{BD}). Note that the angle θ represents the angle that the bar AC makes with horizontal (30° in Figure 1).




Figure 2: Free body diagram on the collar and pulley.

The equilibrium equations are:

$$\sum F_x = 0 \quad -T - R \sin(\theta) + F_{BD} \cos(\alpha) = 0 \quad (2)$$

$$\sum F_y = 0 \quad -T + R \cos(\theta) - F_{BD} \sin(\alpha) = 0 \quad (3)$$

Equations (1) through (3) are three equations in the three unknowns T , R , and F_{BD} .

Computation Because we will be using trigonometric functions, the `$UnitSystem Deg` directive is used to specify that the units of angles must be degree. The inputs include the weight being supported as well as the values of the angles of the two bars.

$W = 5$ [lbf]
 $\alpha = 0$ [deg]
 $\theta = 30$ [deg]

"weight on cable"
 "angle of bar BD"
 "angle of bar AC"

Equations (1) through (3) are directly entered in EES.

$T = W$
 $-T - R \sin(\theta) + F_{BD} \cos(\alpha) = 0$
 $-T - F_{BD} \sin(\alpha) + R \cos(\theta) = 0$

"free body diagram on weight"
 "sum of forces in x-direction"
 "sum of forces in y-direction"

Solving leads to $R = 5.774$ lbf and $F_{BD} = 7.887$ lbf.

Part (b)

Computation Because we have a computer solution to the problem we can directly explore the effect of the angle α on the force F_{BD} using a Parametric Table. We will place the assignment of the variable alpha in a \$IfNot ... \$EndIf directive


```

$IfNot Parametric
alpha = 0 [deg] "angle of bar BD"
$EndIf
  
```

and create a Parametric Table which contains the variables alpha and F_{BD} , as shown in Figure 3(a). Right click on the column for alpha and select Alter Values in order to vary α from -50° to 50° , as shown. Select Solve Table from the Calculate menu to determine the value of F_{BD} corresponding to the value of α in each row. These data are plotted to show F_{BD} as a function of α in Figure 3(b). Notice that the value of F_{BD} reaches a minimum at $\alpha = -30^\circ$.

Run	α [deg]	F_{BD} [lbf]
Run 1	-50	7.268
Run 2	-38.89	6.913
Run 3	-27.78	6.835
Run 4	-16.67	7.019
Run 5	-5.556	7.503
Run 6	5.556	8.395
Run 7	16.67	9.953
Run 8	27.78	12.81
Run 9	38.89	18.96
Run 10	50	39.33

(a)



(b)

Figure 3. (a) Parametric Table in which the variable alpha is set and varied from -50° to 50° and the value of the variable F_{BD} is calculated in each row. (b) Plot showing F_{BD} as a function of α .

Discussion & Verification We did not use vectors to solve this problem but we could have. There are four forces acting on the collar, as shown in Figure 4. These include two related to the tension

which act along the negative x and negative y directions (\vec{F}_{BE} and \vec{W} , respectively), the reaction force (\vec{R}), and the force from the bar (\vec{F}_{BD}).




Figure 4. Vector forces acting on the collar.

The vector sum of these four forces must be the zeros vector.

$$\vec{F}_{BE} + \vec{W} + \vec{R} + \vec{F}_{BD} = \vec{0} \quad (3)$$

This single vector equation corresponds to two scalar equations for this 2-D problem allowing us to solve for the unknown magnitudes of the forces \vec{R} and \vec{F}_{BD} . The solution using vectors is shown in Figure 5; note that the variable **VectorZeros** corresponds to the zero vector, $\vec{0}$.




Figure 5. (a) Equations Window showing the vectors defined and used to solve the problem with Eq. (3). (b) Solution Window showing that the solution is the same as what was found in part (a).

Because we have each of the four forces on the collar that are represented as vector variables we can doublecheck our solution by creating a force polygon, as discussed in Section 2.2E. This is shown in Figure 6.




Figure 6. Force polygon for the case where $\alpha = 20^\circ$.

EXAMPLE E3.2  *Retractable Tool Holder*

This example corresponds to Problem 3.80 in the text. The structure shown in Figure 1 is a retractable tool holder that is used in a factory to support a tool at point D . When the tool is to be used, the worker will grasp the tool and apply a downward force to lower it to the position that is needed. When the tool is not in use, the spring causes the tool to retract so that it is out of the way. The spring has $2 \text{ lb}_f/\text{in}$ stiffness and W can be assumed to be vertical.




Figure 1.

If the unstretched length of the spring is 30 in. , and $\theta = \alpha = 60^\circ$ when $W = 0$, determine the value of θ that occurs when $W = 15 \text{ lb}_f$.

SOLUTION

Road Map We will proceed by using EES to compute the value of W that is associated with a specific value of θ . This approach allows us to enter and solve equations one by one without having to enter an entire equation set before we can solve. Once the solution is working for a given value of θ then it is easy to remove that specification and instead specify a value of W and have EES solve for the corresponding value of θ .

Part (a)

Governing Equations and Computation The inputs include the spring constant, k , and unstretched spring length, L_0 , which corresponds to 30 inch according to the problem statement. We will also initially enter a value of θ , although this will eventually be adjusted to achieve the specified value of W .

```
$UnitSystem Degree
$TabStops 3 in
```

```
theta = 70 [deg]           "assumed value of angle"
L_0 = 30 [in]              "unstretched spring length"
k = 2 [lbf/in]            "spring stiffness"
```

Using point B as the origin, the coordinates of the points B and C are (0, 0) inch and (0, 30) inch. The coordinates of point A are computed based on the length of the bar AB and the angle θ .

```
$Vector2D A, B, C
```

```
B = VectorAssign(0,0) [in]           "coordinate of fixed point B"
C = VectorAssign(0,30) [in]         "coordinate of fixed point C"
A = VectorAssignPolar(30 [in],90 [deg] - theta) "coordinate of point A"
```

The position vectors \vec{r}_{AB} and \vec{r}_{AC} are computed using these coordinates

$$\vec{r}_{AB} = B - A, \text{ and} \quad (1)$$

$$\vec{r}_{AC} = C - A. \quad (2)$$

```
$Vector2D r_AB, r_AC
```

```
r_AB = B - A           "position vector from A to B"
r_AC = C - A           "position vector from A to C"
```

A free body diagram on point A is shown in Figure 2 and includes the spring force, \vec{F}_{spring} , the force from the bar, \vec{F}_{AB} , and the force related to the weight of the tool, \vec{W} .




Figure 2. Free body diagram on point A .

The magnitude of the force that the spring exerts on point A is obtained from the product of the spring constant and the extension of the spring, which is the difference between the magnitude of position vector \vec{r}_{AC} and the unstretched spring length

$$F_{spring} = k(|\vec{r}_{AC}| - L_0). \quad (3)$$

The spring force acts in the direction defined by \vec{r}_{AC} and therefore can be written in vector form as:

$$\vec{F}_{spring} = F_{spring} \frac{\vec{r}_{AC}}{|\vec{r}_{AC}|}. \quad (4)$$

`$Vector2D W, F_spring, F_AB`

`magF_spring = k*(VectorMag(r_AC) - L_0) "magnitude of spring force"`

`F_spring = magF_spring*r_AC/VectorMag(r_AC) "spring force"`

To this point, the EES model can be solved after each of these equations are entered. This is convenient for debugging and allows units to be set and checked. Moving forward, this will not be true so we need to be careful that we have a complete set of equations before we enter and solve them.

The force from the weight acts vertically downwards, but its magnitude is unknown

$$\vec{W} = -W \hat{j}. \quad (5)$$

The force from the bar AB acts in the direction defined by \vec{r}_{AB} but has unknown magnitude F_{AB}

$$\vec{F}_{AB} = F_{AB} \frac{\vec{r}_{AB}}{|\vec{r}_{AB}|}. \quad (6)$$

The equilibrium equation for this problem requires that the sum of the forces on point A be equal to zero

$$\vec{W} + \vec{F}_{spring} + \vec{F}_{AB} = \vec{0}. \quad (7)$$

Note that Eq. (7) is a vector equation and therefore implies summing the forces in both the x - and the y -directions. It therefore represents two equations in the two unknown quantities F_{AB} and W .

$W = \text{VectorAssign}(0, -\text{mag}W)$	"weight"
$F_{AB} = \text{mag}F_{AB} * r_{AB} / \text{VectorMag}(r_{AB})$	"force from bar"
$W + F_{\text{spring}} + F_{AB} = \text{VectorZeros}$	"sum of forces must be equal to zero"

Solving provides $W = 7.697 \text{ lbf}$, which is not equal to the specified value of 15 lbf in the problem statement. However, it is now possible to have EES adjust the value of θ in order to achieve the specified value of W . Select Update Guesses from the Calculate menu in order to start the iteration process from a good starting point (which corresponds to the current solution). Then comment out the specified value of the variable theta and add an equation that specifies the required value of the variable magW.

{theta = 70 [deg]}	"assumed value of angle"
magW = 15 [lbf]	"specified value of W"

Solving now leads to $\theta = 83.62^\circ$.

Discussion & Verification We can use the vector plot capability of EES to visualize the solution. For example, we can visualize the free body diagram shown in Figure 2 by making a 2D vector plot that includes the variables W , F_{spring} , and F_{AB} , all plotted starting at the origin. The result is shown in Figure 3.




Figure 3. Free body diagram on point A made using EES' vector plot capability.

Because the model is programmed in a computer we can easily develop a plot showing the vertical force required as a function of θ or as a function of the y -position of point A . First comment out the equation that specifies the value of W and instead generate a Parametric Table that contains θ , W , and A_y . Vary the value of θ from 60° (it's equilibrium position with $W = 0$) to 90° (when the bar is horizontal) by right-clicking on the column header and selecting Alter Values. Then run the

Parametric Table by selecting Solve Table from the Calculate menu. The result is shown in Figure 3. The two plots discussed are shown in Figure 5.

Run	θ [deg]	magW [lbf]	A_y [in]
Run 1	60	3.469E-18	15
Run 2	63.33	2.855	13.46
Run 3	66.67	5.406	11.88
Run 4	70	7.697	10.26
Run 5	73.33	9.762	8.604
Run 6	76.67	11.63	6.918
Run 7	80	13.33	5.209
Run 8	83.33	14.87	3.483
Run 9	86.67	16.28	1.744
Run 10	90	17.57	-7.266E-18

Figure 4. Parametric Table containing the variables of interest.




Figure 5. Magnitude of force \vec{W} as a function of (a) θ and (b) A_y .

EXAMPLE E3.3*Cables, Bars, and Failure Criteria*

This example corresponds to Example 3.7 in the text. The weight W is supported by boom AO and cables AB , AC , and AD , which are parallel to the y , x , and z axes, respectively.

- (a) If cables AB and AC can support maximum forces of 5000 lb_f each, and boom AO can support a maximum compressive force of 8000 lb_f before buckling, determine the largest weight W that can be supported. Assume that AD is sufficiently strong to support W .
- (b) If the supports at points B and C are relocated to points B' and C' , respectively, and $W = 1000 \text{ lb}_f$, determine the forces supported by boom AO and cables AB , AC , and AD .




Figure 1.

SOLUTION

Road Map We will proceed by using EES to compute the forces associated with a specific value of W . This approach allows us to enter and solve equations one by one without having to enter an entire equation set before we can solve. Once the solution is working then it is easy to specify the failure criteria and have EES solve for the corresponding value of W .

Part (a)

Governing Equations We will start by defining the coordinates of points A , B , C , and O based on Figure 1. The position vectors that define the directions of the forces from boom AO and cables AB and AC are computed according to

$$\vec{r}_{AO} = A - O, \quad (1)$$

$$\vec{r}_{AB} = A - B, \text{ and} \quad (2)$$

$$\vec{r}_{AC} = A - C. \quad (3)$$

A free body diagram on point A is shown in Figure 2.




Figure 2. Free body diagram on point *A*.

The force associated with the weight acts vertically downward

$$\vec{W} = -W \hat{k}. \quad (4)$$

The forces associated with the two cables and boom act in the directions defined by their associated position vectors and can be written as

$$\vec{F}_{AB} = F_{AB} \frac{\vec{r}_{AB}}{|\vec{r}_{AB}|}, \quad (5)$$

$$\vec{F}_{AC} = F_{AC} \frac{\vec{r}_{AC}}{|\vec{r}_{AC}|}, \text{ and} \quad (6)$$

$$\vec{F}_{AO} = F_{AO} \frac{\vec{r}_{AO}}{|\vec{r}_{AO}|}. \quad (7)$$

The equilibrium equation for a free body diagram on point *A* requires that the sum of the force vectors be zero

$$\vec{W} + \vec{F}_{AB} + \vec{F}_{AC} + \vec{F}_{AO} = \vec{0} \quad (8)$$

where $\vec{0}$ corresponds to the vector of zeros. Equation (8) corresponds to three equations (one in each coordinate direction) for the three unknown force magnitudes F_{AB} , F_{AC} , and F_{AO} .

Computation The problem can be solved relatively easily using EES with vectors. A value of the weight W is specified in order to solve the problem. This value will eventually be removed in order to solve for the weight corresponding to each of the possible failure criteria.

magW = 500 [lbf]

"weight"

The coordinates of the points *A*, *B*, *C*, and *O* are defined

\$Vector A, B, C, O

A = VectorAssign (9, 20, 12) [ft]	"coordinates of point A"
B = VectorAssign (9, 0, 12) [ft]	"coordinates of point B"
C = VectorAssign (0, 20, 12) [ft]	"coordinates of point C"
O = VectorAssign (0,0,0) [ft]	"coordinates of point O"

and used to define the position vectors \vec{r}_{AO} , \vec{r}_{AB} , and \vec{r}_{AC} .

\$Vector r_AB, r_AC, r_AO	
r_AB = B - A	"position vector from A to B"
r_AC = C - A	"position vector from A to C"
r_AO = O - A	"position vector from A to O"

Finally, Eqs. (4) through (8) are entered.

\$Vector F_AB, F_AO, F_AC, W	
W = -magW* VectorUnit k	"weight"
F_AO = magF_AO*r_AO/VectorMag(r_AO)	"force from bar AO"
F_AB = magF_AB*r_AB/VectorMag(r_AB)	"force from cable AB"
F_AC = magF_AC*r_AC/VectorMag(r_AC)	"force from cable AC"
W+F_AO+F_AB+F_AC = VectorZeros	"equilibrium equation for point A"

The problem can be solved at this point, providing $F_{AC} = 375 \text{ lb}_f$, $F_{AB} = 833.3 \text{ lb}_f$, and $F_{AO} = -1042 \text{ lb}_f$.

In order to determine the maximum load that can be supported we will comment out the assumed value of W

{magW = 500 [lb _f]	"weight"
--------------------------------	----------

and instead specify that cable AB reaches its failure criteria, $F_{AB} = 5000 \text{ lb}_f$

magF_AB = 5000 [lb _f]	"failure criteria for cable AB"
-----------------------------------	---------------------------------

which leads to $W = 3000 \text{ lb}_f$. Repeating this for $F_{AC} = 5000 \text{ lb}_f$ and $F_{AO} = -8000 \text{ lb}_f$ leads to $W = 6670 \text{ lb}_f$ and $W = 3840 \text{ lb}_f$, respectively. Therefore, the largest weight that can be supported is 3000 lb_f .

Part (b)

Governing Equations and Computation The governing equations for part (b) are the same as those for part (a). According to the problem statement, the weight should be set to 1000 lb_f and the coordinates of points B and C should be changed to those of points B' and C' , respectively. The revised EES code is placed in a separate tab, as shown in Figure 3.

```

Equations Window
Main  part b
$TabStops 3 in
magW = 1000 [lbf]                                "weight"

$Vector A, B, C, O
A = VectorAssign(9, 20, 12) [ft]                  "coordinates of point A"
B = VectorAssign(14, 0, 8) [ft]                   "coordinates of point B"
C = VectorAssign(0, 18, 18) [ft]                  "coordinates of point C"
O = VectorAssign(0,0,0) [ft]                      "coordinates of point O"

$Vector r_AB, r_AC, r_AO
r_AB = B - A                                     "position vector from A to B"
r_AC = C - A                                     "position vector from A to C"
r_AO = O - A                                     "position vector from A to O"

$Vector F_AB, F_AO, F_AC, W
W = -magW*VectorUnit_k                           "weight"
F_AO = magF_AO*r_AO/VectorMag(r_AO)              "force from bar AO"
F_AB = magF_AB*r_AB/VectorMag(r_AB)              "force from cable AB"
F_AC = magF_AC*r_AC/VectorMag(r_AC)              "force from cable AC"
W+F_AO+F_AB+F_AC = VectorZeros                  "equilibrium equation for point A"

```

(a)

```

Solution
part b
Unit Settings: SI C kPa kJ mass deg
VECTORS
A = (9, 20, 12) [ft]
B = (14, 0, 8) [ft]
C = (0, 18, 18) [ft]
E_AB = (244.6, -978.3, -195.7) [lbf]
E_AC = (-760.9, -169.1, 507.2) [lbf]
E_AO = (516.3, 1147, 688.4) [lbf]
O = (0, 0, 0) [ft]
r_AB = (5, -20, -4) [ft]
r_AC = (-9, -2, 6) [ft]
r_AO = (-9, -20, -12) [ft]
W = (0, 0, -1000) [lbf]

SCALARS
magF_AB = 1027 [lbf]    magF_AC = 930 [lbf]
magF_AO = -1434 [lbf]  magW = 1000 [lbf]

No unit problems were detected.

```

(b)

Figure 3. (a) Equations Window for part (b) showing the magnitude of the weight set to 1000 lb_f and the coordinates of points *B* and *C* adjusted, and (b) the Solutions Window.

The solution provides $F_{AB} = 1027 \text{ lb}_f$, $F_{AC} = 930 \text{ lb}_f$, and $F_{AO} = -1434 \text{ lb}_f$.

Discussion and Verification We could easily determine the maximum load that could be supported with the adjusted connection points (*B'* and *C'*) by commenting out the specified value of *W* and setting the failure criteria in each of the members, as we did in part (a). The result is that cable *AB* reaches its failure criteria first at a weight of 4868 lb_f.

One way to view the failure criteria is to make a plot of the magnitude of the force in each of the structural members as a function of the weight; this is done using a Parametric Table in EES and shown in Figure 4.




Figure 4. Magnitude of the force supported in each member as a function of the weight.

The failure criteria for the cables (5000 lb_f) and the boom (-8000 lb_f) are also shown in Figure 4 and it is clear that the force in cable AB hits its failure criteria first.

3.3E The \$VectorPlot Directive

Section 2.2E describes how to create a vector plot using the Vector Plot dialog. This requires that you one by one add vectors to the plot until you are done. Modifying and adjusting the vectors that are plotted is difficult. The \$VectorPlot directive is available if you are using an Academic Professional license of EES and allows you to quickly and programmatically generate a vector plot based on the results of an EES program. The format of the \$VectorPlot directive (and any other directive) can be quickly determined using the Directive Information Dialog which is accessed by selecting Directive Info from the Options menu, as shown in Figure 3.17.




Figure 3.17. Directive Information Dialog.

Navigate to the category of interest in the left box (e.g., Plots) and then select the directive of interest (e.g., `$VectorPlot`) in order to obtain a sample call to that directive. The protocol for the `$VectorPlot` directive is shown below:

```
$VectorPlot Name='MyPlot' Vector1 : origin1/color1 Vector2 :origin2/color2 ...
```

The string 'MyPlot' is the name of the plot that is generated. **Vector1** is the name of the first vector variable to be plotted, origin1 is its origin, and color1 is its color. You can add additional vectors to this list (e.g., **Vector2** with origin2 and color2). The origin can either be the point 0,0,0 (indicated by 0), the name of another vector, or the head of the last vector plotted (indicated by !). Each time that the EES code with the `$VectorPlot` directive is run the plot is reconstructed with the latest values for the vectors involved.

For example, we can automatically plot two vectors and their cross product. The EES code below declares three vectors variables (**A**, **B**, and **C**) and then specifies the components of **A** and **B** and defines **C** to be their cross product.

```
$Vector A, B, C
A = VectorAssign(-1,2,-1)
B = VectorAssign(1,0.5,0.5)
C = VectorCross(A,B)
```

The `$VectorPlot` directive below

```
$VectorPlot Name='ABC' A:0/black B:0/black C:0/red
```

makes a vector plot called 'ABC' that includes all three vectors emanating from the origin, as shown in Figure 3.18(a).




Figure 3.18. (a) Vector plot 'ABC' with the vectors drawn starting at the origin and (b) Vector plot 'ABC2' with the vectors drawn head to tail.

The `$VectorPlot` directive below

```
$VectorPlot Name='ABC2' A:0/black B:!/black C:!/red
```

makes a vector plot called 'ABC2' that shows the vectors drawn head to tail, as shown in Figure 3.18(b).

The same process is used for two-dimensional vectors but the `$VectorPlot2D` directive is used instead of the `$VectorPlot` directive.

EXAMPLE E3.4 Summing Forces in a Direction Other Than x , y , or z

This example corresponds to Example 3.8 in the text. Bar AB is straight and is fixed in space. Spring CD has 3 N/mm stiffness and 200 mm unstretched length. If there is no friction between collar C and bar AB , determine

- The weight W of the collar that produces the equilibrium configuration shown.
- The reaction between the collar and the bar AB .




Figure 1.

SOLUTION

Road Map We will specify the given information which includes the characteristics of the spring and the coordinates of points A , B , C , and D . This is sufficient to define the directions of the rod and the spring. The direction of the weight force is vertical downwards. The sum of the forces of the weight and the spring *in the direction of the rod* must be zero if the rod is frictionless. The sum of the forces associated with the reaction, the spring, and the weight must be zero.

Part (a)

Governing Equations We will start by defining the coordinates of points A , B , and D based on Figure 1. The position vector that defines the bar is determined from

$$\vec{r}_{AB} = B - A. \quad (1)$$

Point C is displaced from point A by a distance $r_{AC} = 240$ mm in the direction defined by \vec{r}_{AB}

$$C = A + r_{AC} \frac{\vec{r}_{AB}}{|\vec{r}_{AB}|}. \quad (2)$$

The position vector corresponding to the spring is

$$\vec{r}_{CD} = D - C. \quad (3)$$

The magnitude of the spring force is given by

$$F_{CD} = k(|\vec{r}_{CD}| - L_0) \quad (4)$$

where L_0 is the unstretched length of the spring. The spring force is in the direction defined by \vec{r}_{CD}

$$\vec{F}_{CD} = F_{CD} \frac{\vec{r}_{CD}}{|\vec{r}_{CD}|}. \quad (5)$$

The force due to the weight is in the vertical downwards direction

$$\vec{W} = -W \hat{k}. \quad (6)$$

The sum of the spring force and weight must be zero in the direction of the rod,

$$\vec{F}_{CD} \cdot \frac{\vec{r}_{AB}}{|\vec{r}_{AB}|} + \vec{W} \cdot \frac{\vec{r}_{AB}}{|\vec{r}_{AB}|} = 0 \quad (7)$$

or, multiplying through by $|\vec{r}_{AB}|$

$$\vec{F}_{CD} \cdot \vec{r}_{AB} + \vec{W} \cdot \vec{r}_{AB} = 0 \quad (8)$$

Equation (8) is a scalar equation that can be solved for the unknown value of W .

Computation The problem inputs include the spring stiffness, unstretched length, and distance between the end of the bar (A) and the location of the collar (point C). Also, the coordinates of points A , B , and D are given.

$k = 3$ [N/mm]	"spring stiffness"
$L_0 = 200$ [mm]	"spring unstretched length"
$\text{magr_AC} = 240$ [mm]	"distance from A to C"
\$Vector A, B, C, D	
$\mathbf{A} = \mathbf{VectorAssign}(0, 60, 240)$ [mm]	"coordinates of point A"
$\mathbf{B} = \mathbf{VectorAssign}(120, 300, 0)$ [mm]	"coordinates of point B"
$\mathbf{D} = \mathbf{VectorAssign}(120, 0, 280)$ [mm]	"coordinates of point D"

Equations (1) through (3) are entered to locate point C and define \vec{r}_{CD} .

\$Vector r_AB, r_AC, r_CD

$$\begin{aligned} \mathbf{r}_{AB} &= \mathbf{B} - \mathbf{A} && \text{"position vector } r_{AB} \text{ - corresponding to the bar"} \\ \mathbf{C} &= \mathbf{A} + \text{magr}_{AC} \mathbf{r}_{AB} / \text{VectorMag}(\mathbf{r}_{AB}) && \text{"coordinates of point C"} \\ \mathbf{r}_{CD} &= \mathbf{D} - \mathbf{C} && \text{"position vector } r_{CD} \text{ - spring} \end{aligned}$$

Equations (4) and (5) define the force from the spring.

$$\begin{aligned} \text{magF}_{CD} &= k * (\text{VectorMag}(\mathbf{r}_{CD}) - L_0) && \text{"magnitude of spring force"} \\ \mathbf{F}_{CD} &= \text{magF}_{CD} * \mathbf{r}_{CD} / \text{VectorMag}(\mathbf{r}_{CD}) && \text{"spring force"} \end{aligned}$$

Finally, Eqs. (6) and (8) can be solved to determine W .

$$\begin{aligned} \mathbf{W} &= -\text{mag}W * \text{VectorUnit } \mathbf{k} && \text{"weight force"} \\ \text{VectorDot}(\mathbf{F}_{CD}, \mathbf{r}_{AB}) + \text{VectorDot}(\mathbf{W}, \mathbf{r}_{AB}) &= 0 && \text{"summation of forces in the direction of the bar"} \end{aligned}$$

Solving provides $W = 400$ N.

Part (b)

Governing Equations and Computation A free body diagram on the collar requires that the sum of the spring force, weight and reaction force must be equal to zero.

$$\vec{F}_{CD} + \vec{W} + \vec{R} = 0 \quad (9)$$

$$\begin{aligned} \mathbf{R} + \mathbf{F}_{CD} + \mathbf{W} &= \text{VectorZeros} && \text{"reaction force on bar"} \\ \text{magR} &= \text{VectorMag}(\mathbf{R}) && \text{"magnitude of reaction force on bar"} \end{aligned}$$

Solving provides $R = 300$ N.

Discussion and Verification The dot product of the reaction force and the position vector that defined the bar should be zero if the reaction is perpendicular to the bar. This is easily checked.

$$\text{check} = \text{VectorDot}(\mathbf{R}, \mathbf{r}_{AB}) \quad \text{"check that reaction force on bar is perpendicular to bar"}$$

Solving shows that check is zero (to within numerical precision).

We can setup a Parametric Table that changes the value of the distance from the end of the bar to the collar, r_{AC} , and examines the resulting value of W . The results are used to make the plot shown in Figure 2. Notice that at $r_{AC} = 240$ mm we get $W = 300$ N which is the answer to part (a). At approximately $r_{AC} = 120$ mm we see that $W = 0$ N which must correspond to the equilibrium position of a weightless collar. Below $r_{AC} = 120$ mm the weight would have to be negative which is not physical.




Figure 3. Weight of the collar as a function of the distance from the collar to the end of the bar.

We can use this problem as an opportunity to demonstrate the use of the `VectorPlot` directive. Let's build a vector plot that includes the position vector \vec{r}_{AB} starting at point A as well as the three forces \vec{W} , \vec{F}_{CD} , and \vec{R} all starting at point C . The result is shown in Figure 4 and helps visualize the forces involved on the collar as well as their relationship to the bar.




Figure 4. Vector plot showing the bar and the three forces acting on the collar.